
Abstract 

A powerful and essential PROC SQL programming technique that all SAS® users should understand, and be comfortable performing, 

is the process of joining (or combining) two or more tables of data. This paper and presentation describes and illustrates the join 

process, including what a join is, exploration of one-to-one, one-to-many, and many-to-many data relationships, identifying a 

primary key (or unique identifier), special preparation requirements for each table being specified in a join, and popular join 

techniques available to SAS® users. To demonstrate the power of the join process, examples of conventional (symmetrical 

matching) and unconventional (asymmetrical left, right and full matching) using PROC SQL join programming techniques are 

illustrated. 

 

Introduction 

The SQL procedure is a wonderful tool for querying and subsetting data; restructuring data by constructing case expressions; 

constructing and using virtual tables known as a view; access information from Dictionary tables; and joining two or more 

tables to explore data relationships. Occasionally, an application problem comes along where the SQL procedure is either better 

suited or easier to use than other more conventional DATA and/or PROC step methods. As each situation presents itself, PROC 

SQL should be examined to see if its use is warranted for the task at hand. 

 

Why Join Anyway? 

Relational database systems continue to grow in popularity, the need to access normalized data stored in separate tables 

becomes increasingly important. A join of two or more tables provides a means of gathering and manipulating data in a single 

SELECT statement. Joins are specified on a minimum of two tables at a time, where a column from each table is used to connect 

the two tables. Connecting columns should have "like" values and the same datatype attributes since the join's success is 

dependent on these values. By relating matching values in key columns in one table with key columns in two or more tables, 

information can be retrieved as if the data were stored in one huge file. Consequently, the process of joining data from two or 

more tables can provide new and exciting insights into data relationships. 

 

What Happens During a Join? 

Joins are processed in two distinct phases. The first phase determines the names of the tables referenced in the FROM clause. 

An internal virtual table, known as a Cartesian product, is created resulting in each row in the first table being combined with 

each row in the second table, and so forth. Due to its size, the Cartesian product is managed by the SAS software. The second 

phase of every join processes the WHERE clause, when present. 

 

Data Relationships and Cardinality 

Tables of data and the relationships of data in a Relational Database Management System (RDBMS) environment represent an 

important topic among data base and SAS users. When it comes to understanding tables, data and the relationship of data, the 

term, Cardinality, is often used to associate the uniqueness of data values contained in a table’s columns. Many working in an 

RDBMS environment associate a table that contains columns possessing a high concentration of unique values as having high 

cardinality. Conversely, a table with columns possessing a low concentration of unique values (or many “repeats” of values) is 

referred to as having low cardinality. 

 

Cardinality is also associated with the relationship between two, or more, tables of data in a RDBMS. Cardinality between tables 

can be represented as one-to-one, one-to-many, and many-to-many data relationships. Typically, table relationships are 

identified and processed using a “key” column in each table. 

 One-to-one – Each row in one table is linked (or related) to a single row in another table using a “key” column. 

 

 One-to-many – Each row in one table is linked (or related) to one, or more, rows in another table using a “key” 

column. 

 

 Many-to-many – One, or more, rows in one table is linked (or related) to one, or more, rows in another table using a 

“key” column. 

 

 

The SQL Optimizer and Join Algorithms 

The RDBMS environment and SQL processing is constantly being monitored to maximize efficient processing. The monitoring 

process is performed by the SQL Optimizer. Classified as a cost-based optimizer, the SQL Optimizer analyzes the available query 

execution plans available to it, determines the costs associated with each execution plan, and selects the execution plan with 

the lowest cost. To help achieve optimal processing efficiencies , the SQL optimizer has four algorithms to choose from when 

performing a join: 

 

 Nested Loop – A nested loop join algorithm may be selected by the SQL optimizer when processing small tables of 

data where one table is considerably smaller than the other table, the join condition does not contain an equality 

condition, first row matching is optimized, or using a sort-merge or hash join has been eliminated from consideration. 

 

 Sort-Merge – A sort-merge join algorithm may be selected by the SQL optimizer when the tables to be joined are 

small to medium size and a nested loop, index and hash join algorithm have been eliminated from consideration. 

 

 Index – An index join algorithm may be selected by the SQL optimizer when indexes created on the columns 

participating in the join relationship are inclined to improve performance. 

 

 Hash – A hash join algorithm may be selected by the SQL optimizer when sufficient memory is available to the system, 

and the BUFFERSIZE option is large enough to store at least the smaller of the tables into memory. 

 
 

Join and the Cartesian Product 

When a WHERE clause is omitted from a join query, all possible combinations of rows in each table is produced. This form of 

join is known as a Cartesian Product. Say, for example, you join two tables where one table has 1,000 rows and the other table 

has 20 rows without a WHERE-clause – the resulting Cartesian product would have 20,000 rows. The real importance of a 

Cartesian product join is to serve as a base (or internal representation) for all joins, as illustrated below. 

 

Authors 

AuthorID 

AuthorName 

 

Books 

BookID 

BookTitle 

AuthorID 

 

 

The next example illustrates a Cartesian product join using a SELECT query without a WHERE clause. Note: The AuthorID column 

is displayed twice – once from the Authors table and the second time from the Books table. 

 

 
PROC SQL ; 

 SELECT * 

  FROM mydata.Authors, 

       mydata.Books(keep=BookID BookTitle AuthorID) ; 

QUIT ; 

 

PROC SQL Join Query 

.  .  .   .  .  .  .  .  . 

 

 
 

 

  

PROC SQL Join Query Results 

Conventional Join with Two Tables and a Where Clause 

A conventional join of two or more tables, uses a WHERE- or HAVING-clause to produce a result set of “matched” rows. Joining 

two tables together is a relatively easy process in SQL, as is illustrated in the following diagram. As can be seen, the Authors and 

Books table are joined together using the “key” AuthorID. 

 

 
 

 

 

Authors 

AuthorID 

    AuthorName 

 

Books 

    BookID 

    BookTitle 

AuthorID 

 

 

The next example illustrates a join of the contents of the tables, Authors and Books, using the “key”, AuthorID in a WHERE-

clause. When the value of AuthorID is equal in both tables, the rows are combined together. Note: The AuthorID column is 

displayed twice in the results – once from the Authors table and a second time from the Books table. 

 

 
PROC SQL ; 

 SELECT * 

  FROM mydata.Authors(drop=AuthorBio), 

       mydata.Books(keep=BookID BookTitle AuthorID) 

   WHERE Authors.AuthorID = Books.AuthorID ; 

QUIT ; 

 

 

  

PROC SQL Join Query 

Join and Table Aliases 

Table aliases provide a "short-cut" way to reference two or more tables in a join operation. Table aliases are specified in the 
FROM-clause so columns can be referenced in a minimal number of keystrokes in the SELECT and WHERE-clause. To illustrate 
how table aliases in a join works, a two-table join is linked together as shown in the following diagram. 
 

Authors 

AuthorID 

    AuthorName 

 

Books 

    BookID 

    BookTitle 

AuthorID 

 

 

The following SQL code illustrates a join on two tables with TITLE specified as the connecting column. The table aliases are 
specified in the SELECT statement as qualified names, the FROM clause, and the WHERE clause. 
 

 
PROC SQL ; 

 SELECT A.AuthorID, AuthorName, BookID, BookTitle 

  FROM mydata.Authors(drop=AuthorBio) AS A, 

       mydata.Books(keep=BookID BookTitle AuthorID) AS B 

   WHERE A.AuthorID = B.AuthorID ; 

QUIT ; 

 
 

PROC SQL Join Query Results 

PROC SQL Join Query 



 One-to-One, One-to-Many, and Many-to-Many Using PROC SQL, continued          SCSUG 2017 

 
Page 6 

 

 

 
 
 

Join with Three Tables 

In an earlier example, a two-table join was shown using the Authors and the Books tables. To illustrate what book was 
purchased as well as the date a book was purchased, a three table join will be constructed. To accomplish this, three different 
tables will be accessed: Authors, Books, and Customers. 
 
A join with three tables adheres to the same rules as a two-table join. Each table needs to be listed in the FROM clause with the 
appropriate subsetting (or matching) restrictions specified in a WHERE clause. The diagram, below, illustrates the table 
relationships along with the keys used in a three table join. 
 

Customers  Books 

    CustomerID  BookID 

BookID      BookTitle 

    PurchaseDate  AuthorID 

 
 
 

 
 
The next example references a three table join with BookID acting as the connecting column for the Customers and Books 
tables, and AuthorID acting as the connecting column for the Authors and Books tables. 
 

Authors 

AuthorID 

    AuthorName 

PROC SQL Join Query Results 

PROC SQL ; 

 SELECT A.AuthorID, AuthorName, B.BookID, BookTitle, PurchaseDate 

  FROM mydata.Authors(drop=AuthorBio) AS A, 

       mydata.Books(keep=BookID BookTitle AuthorID) AS B, 

       mydata.Customers AS C 

   WHERE A.AuthorID = B.AuthorID 

     AND B.BookID = C.BookID ; 

QUIT ; 

 

 

 
 
 

Join with Four Tables 

A join with four tables adheres to the same rules as a two- and three-table join. Each table needs to be listed in the FROM 
clause with the appropriate subsetting (or matching) restrictions specified in a WHERE clause. The diagram, below, illustrates 
the table relationships along with the keys used in a four table join. 
 

Customers  Books 

    CustomerID  BookID 

BookID      BookTitle 

    PurchaseDate  AuthorID 

  BookCategoryID 

 
 
 

BookCategories 

BookCategoryID 

    BookCategory 

    Notes 

 

 

 

Authors 

AuthorID 

    AuthorName 

PROC SQL Join Query 

PROC SQL Join Query Results 

In the next example, a four-table join is constructed for identifying the book category along with the author name, book title, 

hardcopy price, and customer purchase date from the Authors, Books, Customers, and BookCategories tables. To accomplish 

this specific join, the AuthorID from the Authors and Books tables are matched, the BookID from the Books and Customers 

tables are matched, and the CategoryID from the Books and BookCategories tables are matched in a WHERE-clause. Finally, the 

results are arranged and displayed in ascending order. 

 

 
proc sql ; 

 select AuthorName, BookTitle, HardcoverPrice, PurchaseDate, BookCategory 

  from mydata.Authors, 

       mydata.Books, 

       mydata.Customers, 

       mydata.BookCategories 

   where Authors.AuthorID = Books.AuthorID 

     AND Books.BookID = Customers.BookID 

     AND Books.BookCategoryID = BookCategories.BookCategoryID 

    order by PurchaseDate ; 

quit ; 

 

 

 
 
 

Outer Joins 

A conventional join is a process of relating rows in one table with rows in another symmetrically. But occasionally, you may 
want to select the matching rows along with the preservation of unmatched rows from one, both or all tables. This approach is 
sometimes referred to as an asymmetric (or unconventional) join because its basic purpose is to select the matching rows from 
all tables, and to capture the rows without a match row from all tables. This type of join construct serves a significant purpose 
when working with tables of data and is referred to as an outer join construct. 
 
 

 
 
 
 

PROC SQL Join Query 

PROC SQL Join Query Results 

There are syntax and operational differences between inner (natural) and outer joins. The obvious difference between an outer 
and inner join is the way the syntax is constructed. Outer joins use keywords such as LEFT JOIN, RIGHT JOIN, and FULL JOIN, and 
has the WHERE clause replaced with an ON clause. These distinctions help identify outer joins from inner joins. But, there are 
important operational differences as well. 
 
Unlike an inner join, the maximum number of tables that can be specified in an outer join construct is two. Similar to an inner 
join, an outer join relates rows in both tables. But this is where the similarities end since the resulting set of data also includes 
rows with no related rows from one or both of the tables. This special handling of “matched” and “unmatched” rows of data is 
what differentiates an outer join from an inner join. Essentially the resulting set of data from an outer join process contains 
rows that “match” the ON-clause plus any “unmatched” rows from the left, right, or both tables. 
 
An outer join can accomplish a variety of tasks that would require a great deal of effort using other methods. This is not to say 
that a process similar to an outer join cannot be programmed – it would probably just require more work. Let’s take a look at a 
few hypothetical tasks that are possible using outer joins: 
 

 List all customer accounts with rentals during the month, including customer accounts with no purchase activity. 

 Compute the number of rentals placed by each customer, including customers who have not rented. 

 Identify movie renters who rented a movie last month, and those who did not. 

 Identify a list of movie titles that an actor/actress appeared in, and movies they did not appear in. 

 
Finally, specifying a left or right outer join is a matter of choice. Simply put, the only difference between a left and right join is 
the order of the tables used to relate rows of data. As such, the two types of outer joins can be specified based on convenience. 
 

Left Outer Joins 
Outer joins process data relationships from two tables differently than inner joins. In the next example a left outer join is 
constructed to select the “matched” AuthorIDs from both the Authors and Books tables, plus all the “unmatched” rows from 
the Authors table. The result contains all rows for which the SQL expression, referenced in the ON clause, matches the rows 
from the left table (Authors) that did not match any row in the right (Books) table. Essentially any “unmatched” rows from the 
left table are preserved and displayed as they appear in the table itself. 
 
 

 
 
 

 
PROC SQL ; 

 SELECT Authors.AuthorID, BookTitle, HardcoverPrice format=Dollar8.2 

  FROM mydata.Authors 

    LEFT JOIN 

       mydata.Books 

   ON Authors.AuthorID = Books.AuthorID ; 

QUIT ; 

PROC SQL Join Query 

.  .  .   .  .  .  .  .  . 

 

 
 

 

Right Outer Joins 
The next example illustrates the result of using a right outer join to identify and match AuthorIDs from the Authors and Books 
tables. The result contains all rows for which the SQL expression, referenced in the ON clause, matches the rows from the right 
table (Books) that did not match any row in the left (Authors) table. 

 

PROC SQL Join Query Results 

PROC SQL ; 

 SELECT Authors.AuthorID, BookTitle, HardcoverPrice format=Dollar8.2 

  FROM mydata.Authors 

    RIGHT JOIN 

       mydata.Books 

   ON Authors.AuthorID = Books.AuthorID ; 

QUIT ; 

 

 

 
 

Full Outer Joins 
A full outer join essentially represents the result of a left outer join and a right outer join. The result of a full outer join can be 
sizeable because it contains all “matched” and “unmatched” rows from both the left and right table. In the next example a full 
outer join query is constructed that selects the AuthorID, BookTitle, and HardcoverPrice from the Authors and Books tables. 
The result contains all rows that satisfy the SQL expression, referenced in the ON clause, by matching the rows from the right 
table (Books) that did not match any row in the left (Authors) table. 
 

 
PROC SQL ; 

 SELECT Authors.AuthorID, BookTitle, HardcoverPrice format=Dollar8.2 

  FROM mydata.Authors 

    FULL JOIN 

       mydata.Books 

   ON Authors.AuthorID = Books.AuthorID ; 

QUIT ; 

PROC SQL Join Query 

PROC SQL Join Query Results 

PROC SQL Join Query 



 One-to-One, One-to-Many, and Many-to-Many Using PROC SQL, continued          SCSUG 2017 

 
Page 12 

Displaying a Query’s Execution Plan 

The SQL procedure offers users with a very important option to help understand a query’s execution plan. Using a _METHOD 
option in a PROC SQL statement, users can display on the SAS Log the hierarchy of processing incurred by the SQL Optimizer. A 
list of the _METHOD codes, along with their descriptions, is displayed in the table, below. 
 

Code Description 

SQXCRTA Create table as Select 

SQXSLCT Select 

SQXJSL Step loop join (Cartesian) 

SQXJM Merge join 

SQXJNDX Index join 

SQXJHSH Hash join 

SQXSORT Sort 

SQXSRC Source rows from table 

SQXFIL Filter rows 

SQXSUMG Summary stats with GROUP BY 

SQXSUMN Summary stats with no GROUP BY 

 
 
In the next example a _METHOD option is specified to display the three-table join query’s execution plan. 
 

 
PROC SQL _METHOD ; 

 SELECT A.AuthorID, AuthorName, B.BookID, BookTitle, PurchaseDate 

  FROM mydata.Authors(drop=AuthorBio) AS A, 

       mydata.Books(keep=BookID BookTitle AuthorID) AS B, 

       mydata.Customers AS C 

   WHERE A.AuthorID = B.AuthorID 

     AND B.BookID = C.BookID ; 

QUIT ; 

 

 
 

NOTE: SQL execution methods chosen are: 

  

       sqxslct 

           sqxjm 

               sqxjm 

                   sqxsrc( MYDATA.BOOKS(alias = B) ) 

                   sqxsrc( MYDATA.CUSTOMERS(alias = C) ) 

               sqxsrc( MYDATA.AUTHORS(alias = A) ) 

 
 

 

Specifying the BUFFERSIZE= Option 

If surplus virtual (or real) memory is available, in-memory processing may be utilized by the SQL Optimizer to improve the 

performance associated with a query’s matching, aggregation, and intersection processing. One technique uses a BUFFERSIZE= 

option to specify the internal buffer page size for paged memory (up to the size of the MEMSIZE= value) processing including 

performing in-memory, and faster, operations. Since memory speeds are generally faster (nanoseconds) in comparison to 

secondary storage (milliseconds), a query’s performance can be improved. In the next example, a BUFFERSIZE=1024000 option 

(other values can be specified based on the amount of available memory) is specified to tell the SQL Optimizer the amount of 

memory that is available for in-memory processing. 

 

PROC SQL Join Query 

PROC SQL Join Query Results 

PROC SQL _METHOD BUFFERSIZE=1024000 ; 

 SELECT A.AuthorID, AuthorName, B.BookID, BookTitle, PurchaseDate 

  FROM mydata.Authors(drop=AuthorBio) AS A, 

       mydata.Books(keep=BookID BookTitle AuthorID) AS B, 

       mydata.Customers AS C 

   WHERE A.AuthorID = B.AuthorID 

     AND B.BookID = C.BookID ; 

QUIT ; 

 

 
 

NOTE: SQL execution methods chosen are: 

  

       sqxslct 

           sqxjhsh 

               sqxjhsh 

                   sqxsrc( MYDATA.BOOKS(alias = B) ) 

                   sqxsrc( MYDATA.CUSTOMERS(alias = C) ) 

               sqxsrc( MYDATA.AUTHORS(alias = A) ) 

 

 

 

Conclusion 

The SQL procedure is a tool for SAS users to explore and use in a variety of application situations, including powerful and 

essential join constructs using PROC SQL to combine two or more tables of data together. This paper and presentation described and 

illustrated the join process, including what a join is, exploration of one-to-one, one-to-many, and many-to-many data relationships, 

identifying a primary key (or unique identifier), and popular join techniques available to SAS users. The application of conventional 

(or symmetrical matching) and unconventional (or asymmetrical left, right and full matching) join queries were also illustrated. 

Tables Used in Examples 

The examples used throughout this paper utilize a database of five tables (A relational database is a collection of tables.) The 

tables and data correspond to a selection of SAS authors, book categories, book formats, books and customers. The Authors 

table consists of two columns: AuthorID and AuthorName, both of which are defined as character columns, illustrated below. 

 

 

  
 

The BookCategories table consists of three columns: BookCategoryID, BookCategory, and Notes, all of which are defined as 

character columns. The physical data stored in the BookCategories table is illustrated below. 

 

 

 
 

 

 

Authors Table 

BookCategories Table 

The BookFormats table consists of three columns: BookCategoryID, BookCategory, and Notes, all of which are defined as 

character columns. The physical data stored in the BookCategories table is illustrated below. 

 

 

 
 

 

The Books table consists of fifteen columns: BookID, BookTitle, AuthorID, BookCategoryID, Publisher, Edition, CopyrightDate, 

FormatID, HardcopyISBN, ePubISBN, MobiISBN, PDFISBN, PageCount, HardcoverPrice, and BookHyperlink, all of which are 

defined as character columns except PageCount and HardcoverPrice are defines as numeric columns. The physical data stored 

in the Books table is illustrated below. 

 

 

 

BookFormats Table 

Books Table 

The Customers table consists of three columns: CustomerID, BookID, and PurchaseDate, where CustomerID and BookID are 

defined as character columns and PurchaseDate is defined as a numeric column. The physical data stored in the Customers 

table is illustrated below. 

 

 

 
 
 
  

Customers Table 

